

9T10A DIAGNOSTIC TAPE PROGRAM
PROGRAM TO WRITE DIAGNOSTICS ON TAPE

A. PURPOSE OF PROGRAM

1. TO PROVIDE A PROGRAM TO WRITE DIAGNOSTIC PROGRAM CARD DECKS ON TAPE.
2. THIS PROGRAM WILL ALSO MODIFY THE DIAGNOSTIC TAPE, ONCE IT HAS BEEN WRITTEN, FOR ADDITIONS OR CORRECTIONS.

B. METHOD OF PROGRAM

1. GENERATING THE ORIGINAL TAPE.

THIS PROGRAM READS IN PROGRAM DECKS AND WRITES EACH DECK ON TAPE AS A FILE.

THE PROGRAM DECKS SHOULD BE IN THE FOLLOWING ORDER WHEN WRITING A TAPE.

- A. THE FIRST PROGRAM WRITTEN IS AN ADJUSTED 9M06A PROGRAM DECK. THIS IS PLACED AS THE FIRST PROGRAM SINCE ADJUSTMENTS HAVE BEEN MADE SO PROGRAM MAY BE MANUALLY LOADED OR LOADED BY LOAD TAPE BUTTON.
- B. THE SECOND PROGRAM IS 9S01A HIGH AND LOW.
- C. THE THIRD PROGRAM IS 9T11A, WHICH IS THE SEARCH PROGRAM TO BE USED FOR READING IN OTHER PROGRAMS.

THE ABOVE SEQUENCE IS THE BASIC GROUP, WHICH INCLUDES A PROGRAM TO TEST C.P.U. BEFORE RUNNING OTHER PROGRAMS AND A STORAGE PROGRAM TO TEST MEMORY.

- D. THE REMAINING PROGRAMS ON TAPE ARE DIAGNOSTIC PROGRAMS. THEY MAY BE PLACED IN ANY ORDER. HOWEVER A SUGGESTED ORDER IS LISTED IN SECTION II A.

2. MODIFYING AN EXISTING TAPE.

WHEN MODIFYING EXISTING TAPE YOU WILL BE DOING TWO OPERATIONS, EITHER CORRECTIONS OR MAKING ADDITIONS AT END OF TAPE.

3. IF PROGRAM IS USED TO GENERATE A NEW TAPE. SENSE SWITCH 5 WILL

BE UP.

IF PROGRAM IS USED TO MODIFY AN EXISTING DIAGNOSTIC TAPE, SWITCH 5 WILL BE DOWN, WHEN CORRECTIONS ARE MADE A NEW DIAGNOSTIC TAPE IS WRITTEN. WHEN ADDITIONSS ARE MADE, THEY ARE WRITTEN ON THE EXISTING DIAGNOSTIC TAPE.

AREA OF MACHINE REQUIRED

1. MF, CF, CR, DS

- A. TO GENERATE A NEW DIAGNOSTIC TAPE. PLACE NEW TAPE ON CHANNEL B TAPE 1.
- B. TO MODIFY AN EXISTING DIAGNOSTIC TAPE WITH CORRECTIONS. PUT EXISTING TAPE ON CHANNEL A TAPE 1. NEW TAPE TO BE WRITTEN MUST BE ON CHANNEL B TAPE 1.
- C. TO MODIFY A DIAGNOSTIC TAPE BY ADDING NEW DECKS TO REAR OF TAPE, PUT PRESENT TAPE ON CHB TAPE 1.

2. STORAGE LOCATIONS - 00030 - 00400.

D. PROGRAM CONTROL

1. DECK	000	9LD01A LOW END LOADER WITH CARD SELECT
	001-008	9T10A PROGRAM
	009	TRANSFER CARD TRA 00030

THE FOLLOWING CARDS IN THIS DECK ARE TO BE USED AS INSERTION CARDS WHEN WRITING A TAPE. THEIR FUNCTION IS TO CHANGE SENSE SWITCH 6 CONTROL IN A DIAGNOSTIC TO READ TAPE 1 CHA, INSTEAD OF READ CARD READER WHEN PASS COMPLETE.

9M01A	9S03A H+L	9T03A
9M02A	9S04A H+L	9T04A
9M03A	9S05A H+L	9P01A
9M04A	9D01A	9COMA
9M05A	9B01A	
*9M06A SEE NOTE	9B03A	
9S01A H+L	9T01A	
9S02A H+L	9T02A	

NOTE -- 9M06A HAS 24 WORDS PER CARD. CONSEQUENTLY WE CAN NOT USE A CONTROL CARD AS IN THE OTHER DIAGNOSTICS. INSTEAD A CORRECTED CARD WILL BE INCLUDED FOLLOWING THE CONTROL CARDS. THE CARD IS 9M06A 104. INSERT THIS IN THE DECK IN THE PLACE OF THE PERMANENT CARD 104

WHEN WRITING A TAPE.

FOR ANY NEW DIAGNOSTICS RELEASED IN THE FUTURE. THE INSTALLATION WILL HAVE TO MAKE THEIR OWN CONTROL CARDS.

9M06A 000 - SPECIAL LOADER FOR LOADING 24 WORDS PER CARD WHEN PROGRAM IS ON TAPE.

9LD01A - LOW END LOADER WITH RTBA 1 SELECT

9LD02A - HIGH END LOADER WITH RTBA 1 SELECT

9LREW - LOW END REWINDER USED WITH LOW END LOADER

9HREW - HIGH END REWINDER USED WITH HIGH END LOADER

NOTE -- ONLY ONE COPY OF 9LD01, 9LD02, 9LREW, 9HREW HAVE BEEN SUPPLIED. EACH INSTALLATION WILL HAVE TO REPRODUCE THE NUMBER OF DUPLICATES NEEDED FOR EACH DECK PUT ON TAPE.

E. SENSE SWITCH CONTROL

SWT 5 UP WHEN PROGRAM IS USED TO WRITE NEW TAPE.

SWT 5 DN WHEN PROGRAM IS USED TO MODIFY EXISTING TAPE.

ALL OTHER SWITCHES UP.

SENSE LINES -- PROGRAM USES SENSE LINES 1,2 AND 4 DURING PROGRAM EXECUTION.

F. NORMAL STOPS

1. WHEN WRITING A NEW TAPE WITH SWT 5 UP

00151 - NEW TAPE HAS BEEN WRITTEN TAPE 1 CHANNEL B REWINDS. TO USE AS DIAGNOSTIC TAPE, CHANGE TAPE TO CHA TAPE 1.

NOTE -- REMOVE FILE PROTECT RING TO PROTECT TAPE FROM BEING WRITTEN ON.

2. WHEN MODIFYING EXISTING TAPE WITH SWT 5 DN.

00160 - IF INSERTING CORRECTION FILES, ENTER INTO THE ADDRESS PORTION OF THE KEYS. THE TOTAL OCTAL NUMBER OF FILES ON EXISTING TAPE, PUSH START TO CONTINUE.

IF ADDITIONS ARE TO BE ADDED TO END OF TAPE BE SURE ALL KEYS ARE

UP. PUSH START TO CONTINUE.

00163 - WHEN MAKING CORRECTIONS ENTER KEYS RESPECTIVE KEY OR KEYS TO INDICATE THE FILE OR FILES TO BE REPLACED.

WHEN MAKING ADDITIONS TO TAPE, DEPRESS SIGN POSITION KEY AND ENTER INTO THE ADDRESS POSITION OF THE KEYS. THE TOTAL OCTAL NUMBER OF FILES ON PRESENT TAPE. PUSH START TO CONTINUE.

00264 - TAPE 1 CH B SHOULD HAVE REWINDED, FOR ALL CORRECTIONS HAVE BEEN MADE ONTO NEW TAPE.

REMOVE FILE PROTECT RING WHEN COMPLETED

00300 - WHEN ADDING A FILE TO REAR OF PRESENT TAPE, A STOP WILL OCCUR HERE. THE REASON FOR THIS IS THAT THE PROGRAM TRANSERS TO THE ROUTINES FOR WRITING CARDS ONTO TAPE. PUT SWT 5 UP. PUSH START TO CONTINUE.

G. ERROR STOPS

SWT 5 UP.

00066 - BLANK CARD IN MIDDLE OF DECK. RUN OUT CARDS. REMOVE BLANK AND RELOAD LAST 3 CARDS AND REMAINDER OF DECK FOLLOWING BLANK. PRESS START TO CONTINUE.

00101 - HAD 5 TAPE CHECKS IN WRITING. TAPE WOULD NOT WRITE IN 5 TRIES. SUGGEST P.M. BEFORE TRYING TO WRITE TAPE AGAIN.

00120 - TAPE CHECK WHILE READING - PRESS START TRY AGAIN.

00133 - CHECK SUM ERROR IN READING CARD TO TAPE. TAKE CARDS NOT READ INTO READER OUT OF HOPPER. RUN OUT CARDS. RELOAD LAST 4 CARDS THAT RAN OUT OF MACHINE FOLLOWING BY REMAINDER OF DECKS NOT READ IN.

SWT 5 DN.

00177 - NO EOF WHEN SKIPPING OVER FILES BEING CORRECTED. RELOAD DECK AND START OVER.

00215 - TAPE CHECK ON READING TAPE. PRESS START TO TRY AGAIN.

00225 - TAPE CHECK ON WRITING. PRESS START TO TRY AGAIN.

00236 - TAPE CHECK ON READING. PRESS START TO TRY AGAIN.

00250 - CHECK SUM ERROR IN COMPARING NEW COMPUTED CHECK SUM FROM NEW TAPE

TO ITS OLD CHECK SUM. PRESS START TRY AGAIN.

00272 - NO EOF INDICATED WHEN READING OVER FILES TO ADD FILE TO END OF TAPE. PRESS START-TRY AGAIN.

H. COMMENTS

FURTHER DESCRIPTION OF CARD SEQUENCE AND FUNCTIONS.

DIAGNOSTIC DECK WITH 9LD01A LOW END LOADER.

1. 9LD01 LOW END LOADER. - WITH RTBA 1 SELECT
2. DIAGNOSTIC PROGRAM CARDS - EACH INDIVIDUAL CARD WRITES AS A RECORD ON TAPE.
3. CONTROL CARD - TO CHANGE SENSE SWITCH 6 CONTROL IN DIAGNOSTIC FROM READ CARD SELECT TO READ TAPE BINARY CHA TAPE 1.
4. LOW END REWINDER - THIS WILL REWIND TAPE WHEN DIAGNOSTIC BEING SEARCHED FOR HAS BEEN LOCATED WHEN READING DIAGNOSTIC TAPE.
5. TRANSFER CARD - NORMAL TRANSFER CARD WHICH TRANSFERS TO BEGINING OF DIAGNOSTIC LOCATED.
6. TWO BLANK CARDS - TWO BLANK CARDS INDICATES END OF DIAGNOSTIC. THIS INITATES WRITING AN END OF FILE WHEN DEVELOPING A DIAGNOSTIC TAPE.

NOTE -- ONE BLANK CARD WILL NOT WRITE AN EOF, BUT WILL CAUSE AN ERROR CONDITION.

NO BLANK CARDS AT END OF A DIAGNOSTIC DECK IS DESCRIBED UNDER SEQUENTIAL DIAGNOSTIC WRITE UP.

DIAGNOSTIC DECK WITH 9LD02A HIGH END LOADER.

A HIGH END LOADER DECK IS BASICALLY THE SAME AS DESCRIBED IN THE COMMENTS ON LOW END LOADER DECK, EXCEPT FOR THE FOLLOWING.

1. 9LD02 HIGH END LOADER - WITH RTBA 1 SELECT
2. HIGH END REWINDER - THIS WILL REWIND DIAGNOSTIC TAPE WHEN FILE HAS BEEN LOCATED.

1. CARD DECK PREPARATIONS TO WRITE A NEW DIAGNOSTIC TAPE

WHEN WRITING A NEW TAPE.
READY NEW TAPE ON TAPE 1 CHANNEL 9.

1. SEPARATE CARDS 000-009 OF 9T10A DECK. THIS DECK IS THE WRITE PROGRAM WHICH PRECEDES ALL DECKS.
2. TAKE 9M06A DECK.
 - A. PLACE CARD 9M06A 000 IN FRONT OF DECK. THIS IS A SPECIAL LOADER TO LOAD IN THE 9M06A MANUAL LOAD PROGRAM.
 - B. INSERT AND REPLACE CARD 9M06A 104 TO CHANNEL SWT6 CONTROL.
 - C. REMOVE TWO BLANK CARDS AT REAR OF DECK.
3. TAKE 9S01A H+L DECK.
 - A. REPLACE CARD 9SW01AH 000 WITH 9LD02 - TAPE SELECT
 - B. INSERT SWT 6 CONTROL CARD FOR 9S01H BEFORE TRANFER CARD.
 - C. REPLACE 9S01AL 000 WITH 9LD01 - TAPE SELECT
 - D. INSERT SWT 6 CONTROL CARD FOR 9S01L BEFORE TRANSFER CARD.
 - E. REMOVE LAST TWO BLANK CARDS 9S01AL.
 - F. PLACE THE DECK DIRECTLY BEHIND ADJUSTED 9M06 DECK.
4. PLACE 9T11A CARDS 000-009. DIRECTLY BEHIND ADJUSTED 9S01A DECK.

THIS WILL BE THE DIAGNOSTIC TAPE BASIC FILE. THE THREE PROGRAMS WILL BE BROUGHT IN BY DEPRESSING LOAD TAPE BUTTON. THE 9M06 AND 9S01 WILL BE PERFORMED AND THE FILE SEARCH PROGRAM 9T11 WILL STOP AT LOCATION 00036 TO ALLOW SELECTING THE NEXT DESIRED PROGRAM.

5. ALL OTHE RPROGRAMS WILL FOLLOW IN THIS TYPE OF SEQUENCE.

EXAMPLE OF A MF PROGRAM SEQUENCE.

9M01A

CARD 0 - 9LD01A LOW END LOADER WITH TAPE SELECT

CARD 1 -184 DIAGNOSTIC 9M01A
CARD 185 CONTROL CARD CHANGING SWT SELECT
CARD 186 LOW END REW
CARD 187 TRA 30
CARD 188-189 2 BLANK CARDS

EXAMPLE OF A STORAGE PROGRAM SEQUENCE.

9S02A HIGH AND LOW

CARD 0 - 9LD02 HIGH END LOADER WITH TAPE SELECT
CARD 1 -10 DIAGNOSTIC 9S02A H
CARD 11 CONTROL CARD SWT 6
CARD 12 TRANSFER CARD.

NOTE -- NO REWIND CARD IS INSERED HERE BECAUSE OF LOW END STORAGE TEST
FOLLOWING HIGH END. ALSO NO BLANKS ARE HERE SO EOF IS NOT WRITTEN.

CARD 13 9LD01A LOW END LOADER WITH TAPE SELECT
CARD 14 -23 DIAGNOSTIC 9S02A L
CARD 24 CONTROL CARD SWT 6
CARD 25 LOW END REW
CARD 26 TRANSFER CARD
CARD 27 -28 2 BLANK CARDS

IA. DESCRIPTION OF 4 DIAGNOSTICS WHICH REQUIRE SPECIAL ATTENTION BEFORE WRITING
ON TAPE.

1. 9M05A - FLOATING POINT TEST HAS TWO TRANSFER CARDS. THE FIRST TRANSFER
CARD 184 SHOULD BE WRITTEN ON TAPE FOR IT IS THE NORMAL
TRANSFER.

THE SECOND TRANSFER CARD 185 IS FOR TRACING. MANUALLY
TRANSFER INTO TRACING ROUTINE WHEN USING 9M05A FOR TRACING.
REMOVE CARD 185 BEFORE WRITING ON TAPE.

2. 9S04A LOW + 9S05A LOW

CARD 49 IN 9S04AL AND CARD 44 IN 9S05AL HAVE A MINUS TRA.
IN 9 ROW LEFT WHICH IS OF NO USE. REPLACE WITH A BLANK SO THAT
AN EOF MAY BE WRITTEN WHEN WRITING A DIAGNOSTIC TAPE.

3. 9COMA - DO NOT WRITE CARD 095 ON TAPE. THIS CARD BY-PASSES DRUM. TO
BY-PASS DRUM WHEN USING THIS PROGRAM. STORE A TRANSFER 1145
AT LOCATION 1117 MANUALLY. REMOVE CARD 095 WHEN WRITING TAPE.

4. FOR PROGRAMS REQUIRING TEST CARDS, DO NOT WRITE TEST CARDS ON TAPE. TEST CARDS SHOULD BE LOADED IN CARD READER, WHEN DIAGNOSTIC HAS BEEN READ OFF TAPE AND READY FOR TESTING. ONE SUCH PROGRAM IS 9C01A.

IB. SEQUENTIAL DIAGNOSTIC FILE-

IF AN INSTALLATION DESIRES A SEQUENTIAL FILE ON TAPE WITH A GROUP OF DIAGNOSTICS TO BE EXECUTED IN SEQUENCE WITHOUT REWINDING AFTER EACH TEST, THIS CAN BE DONE IF THE REWIND CARD AND BLANKS ARE REMOVED, EXCEPT FOR LAST DECK IN FILE. AN EXAMPLE OF SUCH A FILE USING 9M01, 9M02, 9S01 AND 9T01 IS-

1. LOADER
9M01A
CONTROL CARD
TRA CARD
2. LOADER
9M02A
CONTROL CARD
TRA CARD
3. LOADER
9S01AH
CONTROL CARD
TRA CARD
LOADER
9S01AL
CONTROL CARD
TRA CARD
4. LOADER
9T01A
CONTROL CARD
REWINDER REW TAPE
TRA CARD
2 BLANKS - EOF

II. MACHINE OPERATIONS TO WRITE A NEW DIAGNOSTIC TAPE.

TO WRITE A NEW DIAGNOSTIC TAPE FROM CARD READER. THE CONTROLS ARE AS FOLLOWS.

1. SENSE SWITCH 5 UP.

2. PUT TAPE TO BE WRITTEN ON CHANNEL B TAPE 1.
3. READY CARD READER WITH -

9T10A
9M06A
9S01A
9T11A AND
REMAINING DIAGNOSTIC DECKS.

4. WHEN LAST CARD IS IN CARD READER, PRESS CARD READER START TO RUN OUT CARDS. THIS WILL WRITE THE LAST EOF ON TAPE AND REWIND IT.
5. FINAL PROGRAM HALT AT 00151.
6. REMOVE FILE PROTECT RING AND MOUNT AS CHANNEL A TAPE 1 TO USE.

IIA. SUGGESTED INDEX FOR DIAGNOSTIC TAPE

FILE -FILE-----DIAGNOSTIC-DIAGNOSTIC DECK SET UP AND REMARKS
KEY

1	0	9M06A	SPECIAL LOADER, CONTROL, NO REW, TRA IN LOADER NO BLKS. NO PRINT ID.
BASIC TESTS		9S01AH	HIGH LOADER, CONTROL, NO REW, TRA, NO BLKS.
		9S01AL	LOW LOADER, CONTROL, NO REW, TRA, NO BLKS
		9T11A	LOW LOADER, NO REW, TRA, 2 BLANKS
2	1	9M01A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
3	2	9M02A	HIGH LOADER, CONTROL, HIGH REW, TRA, 2 BLKS
4	3	9M03A	HIGH LOADER, CONTROL, HIGH REW, TRA, 2 BLKS
5	4	9M04A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS NO PRINT ID.
6	5	9M05A	HIGH LOADER, CONTROL HIGH REW. 1 TRA NORMAL CARD, 2 BLANKS, NO PRINT ID.
7	6	9S02AH	HIGH LOADER, CONTROL, NO REW, TRA, NO BLKS
		9S02AL	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
10	7	9S03AH	HIGH LOADER, CONTROL, NO REW, TRA, NO BLKS
		9S03AL	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
11	8	9S04AH	HIGH LOADER, CONTROL, NO REW, TRA, NO BLKS
		9S04AL	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS REMOVE MINUS TRA CARD 49 REPLACE WITH BLANK
12	9	9S05AH	HIGH LOADER, CONTROL, NO REW, TRA, NO BLKS
		9S05AL	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS REMOVE MINUS TRA CARD 44 REPLACE WITH BLANK
13	10	9D01A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
14	11	9B01A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS

15	12	9B03A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
16	13	9T01A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
17	14	9T02A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS NO PRINT ID.
20	15	9T03A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
21	16	9T04A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
22	17	9P01A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
23	18	9COMA	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS DO NOT WRITE CARD 095 BY-PASS DRUM. STORE MANUALLY TRA 1145 AT LOC 1117 TO BY-PASS DRUM.
24	19	9C01A	LOW LOADER, CONTROL, LOW REW, TRA, 2 BLKS
25	20	9V01A	LOW LOAD, LOW REW, TRA 2BLKS NO PRINT ID.
26	21	9S01H	HIGH LOADER, CONTROL, TRA
SEQUENTIAL		9S01L	LOW LOADER, CONTROL, TRA
FILE		9M01A	LOW LOADER, CONTROL, TRA
		9M02A	HIGH LOADER, CONTROL, HIGH REW, TRA 2 BLANKS.

TOTAL PHYSICAL FILES -26 OCTAL

*FILE - INDICATES NUMBER OF PHYSICAL FILES ON TAPE. THIS NUMBER WILL BE USED WHEN MODIFYING TAPE

*FILE KEY - INDICATES KEY TO DEPRESS WHEN USING FILE SEARCH ROUTINE TO LOCATE FILE ON DIAGNOSTIC TAPE.

III. PREPARATIONS FOR MODIFYING THE TAPE EITHER BY CORRECTIONS OR ADDITIONS

TO MODIFY AN EXISTING DIAGNOSTIC TAPE, BY ADDING AN ADDITIONAL FILE AFTER THE LAST FILE ON TAPE, THE FOLLOWING INSTRUCTIONS MUST BE APPLIED.

NOTE -- THIS EXAMPLE WILL BE USED FOR SUGGESTED DIAGNOSTIC TAPE IN THIS WRITE UP.

1. READY CARD READ WITH 9T10A AND NEW DECK OR DECKS.
2. LOAD PRESENT DIAGNOSTIC TAPE ON CHANNEL B TAPE 1. PUT FILE PROTECT RING ON TAPE SO WRITING CAN TAKE PLACE.
3. LOAD CARDS WITH SENSE SWITCH 5 DN.
4. HALT AT LOCATION 160. BE SURE ALL KEYS ARE UP ON CONSOLE. PRESS START
5. HALT AT LOCATION 163. DEPRESS THE SIGN KEY AND TOTAL OCTAL NUMBER OF

PHYSICAL FILES ON TAPE IN ADDRESS PORTIONS OF THE KEYS. THIS EXAMPLE WILL BE 26 OCTAL.

6. TAPE SKIPS OVER 26 FILES. AFTER LAST FILE HAS BEEN READ OVER A HALT OCCURS AT LOCATION 00300. PUT SWT 5 UP AND PRESS START.
7. CARD READER IS SELECTED AND NEW DECKS ARE WRITTEN ON TAPE. WHEN CARD READER HAS READ ALL CARDS PRESS CARD READER START TO RUN OUT CARDS AND WRITE LAST EOF ON TAPE.
8. TAPE REWINDS WHEN FINISHED AND HALT AT LOCATION 00151. TAKE FILE PROTECT RING OFF.

IIIA--TO ADD NEW CORRECTION DECKS BY REPLACING OLD FILE OR FILES ON PRESENT DIAGNOSTIC TAPE. THESE OPERATIONS WRITE A COMPLETE NEW TAPE.

EXAMPLE -- REPLACE PHYSICAL FILES 9M01A FILE 2, 9M04A FILE 5 AND 9S03A H+L FILE 10.

1. LOAD PRESENT DIAGNOSTIC TAPE ON CHANNEL A TAPE 1. LOAD NEW TAPE ON CHANNEL B TAPE 1.
2. READY CARD READER WITH
9T10A
9M01A
9M04A
9S03A
3. HALT AT LOCATION 160. ENTER KEYS IN ADDRESS TOTAL PHYSICAL NUMBER OF FILES ON PRESENT TAPE IN ADDRESS PORTION OF KEYS. THIS EXAMPLE 26 OCTAL IN ADDRESS. PRESS START.
4. HALT AT LOCATION 163. ENTER KEY OR KEYS PHYSICAL FILE BEING REPLACED.

EXAMPLE -- KEY 2 9M01, KEY 5 9M04, KEY 10 9S03 PRESS START.

5. THE PRESENT TAPE WILL WRITE ONTO NEW TAPE. WHEN CORRECTION FILE IS INDICATED THE CARD READER WILL SELECT. WHEN LAST DECK IS THROUGH CARD READER, PRESS CARD READER START TO RUN OUT CARDS AND WRITE LAST EOF.
6. A REWIND OCCURS ON BOTH TAPES WHEN COMPLETE. HALT AT LOCATION 264,
REMOVE FILE PROTECT RING FROM NEW TAPE.

IIIB. WHEN CORRECTIONS AND ADDITIONS ARE TO BE MADE AT THE SAME TIME, THE

PROGRAM WILL MAKE CORRECTIONS AND THEN WRITE THE NEW ADDITIONS.


PROCEDURE AS FOLLOWS.


1. PLACE EXISTING TAPE ON CHA TAPE 1.
NEW TAPE ON CHB TAPE 1.
2. PLACE CORRECTIONS IN CARD READER. THEN FOLLOW INSTRUCTIONS AS
IN SECTION III A.
3. AFTER CORRECTIONS ARE MADE TAPES REWIND AND PROGRAM STOPS AT 0264.
4. NOW PLACE NEW PROGRAMS THAT ARE THE ADDITIONS FOR TAPE, IN THE CARD
READER. SWITCH 5 WILL STILL BE DOWN, TRANSFER MANUALLY TO 0030.
PRESS START. THE REMAINING PROCEDURE IS OUTLINED IN SECTION III.


INDEX OF SECTIONS


- PAGE 1.006 SECTION I - CARD DECK PREPARATIONS TO WRITE A NEW DIAGNOSTIC TAPE
- PAGE 1.007 SECTION IA - DESCRIPTION OF 4 DIAGNOSTICS WHICH REQUIRE SPECIAL ATTENTION BEFORE WRITING A DIAGNOSTIC TAPE.
- PAGE 1.008 SECTION IB - SEQUENTIAL DIAGNOSTIC FILE
- PAGE 1.009 SECTION II - MACHINE OPERATIONS TO WRITE A NEW DIAGNOSTIC TAPE
- PAGE 1.009 SECTION IIA - SUGGESTED INDEX FOR DIAGNOSTIC TAPE
- PAGE 1.010 SECTION III - PREPARATIONS FOR MODIFYING THE TAPE EITHER BY CORRECTIONS OR ADDITIONS
- PAGE 1.011 SECTION IIIA - TO ADD NEW CORRECTION DECKS BY REPLACING OLD FILE OR FILES ON PRESENT DIAGNOSTIC TAPE
- PAGE 1.011 SECTION IIIB - WHEN CORRECTIONS AND ADDITIONS ARE TO BE MADE AT THE SAME TIME

709 DIAGNOSTIC TAPE UTILITY PROGRAM









*PROGRAM TO WRITE A NEW DIAGNOSTIC TAPE WITH SENSE SWITCH 5 UP

*

*PROGRAM TO WRITE MODIFIED AND ADDITIONAL DIAGNOSTICS WITH SENSE SWITCH

*

00030 ORG 24

00030	0760	00	0	00165	SWT 5	
00031	0020	00	0	00033	TRA START	WRITE NEW TAPE FROM CARDS
00032	0020	00	0	00152	TRA TAPE	WRITE MODIFICATION FROM CARDS AND TAPE

*WRITE A NEW TAPE FROM CARDS

*RESET INDICATORS

00033	0772	00	0	02201	START REWB 1	REW TAPE 1
00034	0030	00	0	00035	TEFA *+1	RESET EOF INDICATOR
00035	-0030	00	0	00036	TEFB *+1	
00036	0022	00	0	00037	TRCA *+1	RESET REDUNDANCY INDICATOR
00037	-0022	00	0	00040	TRCB *+1	
00040	0760	00	0	00140	SLF	TURN SENSE LITES OFF

*SELECT CARD READER INITIALLY AT THIS LOCATION

00041	0762	00	0	01321	STRTA RCDA	READ CARD
00042	0540	00	0	00305	RCHA CWA	
00043	0060	00	0	00043	STRTB TCOA *	
00044	0030	00	0	00143	TEFA EOF	TRANSFER ON EOF INDICATES EOF ON CARD READER

*COMPUTE CHECK SUM BEFORE WRITING ON TAPE

00045	0774	00	4	00030	AXT 24,4	FORM CHECK SUM
00046	-0754	00	0	00000	PXD	
00047	0361	00	4	00337	ACL TSA+24,4	
00050	2	00001	4	00047	TIX *-1,4,1	
00051	0602	00	0	00367	SLW TSC	STORE CHECK SUM
00052	-0100	00	0	00062	TNZ AA	CARD NOT BLANK

*TWO BLANK CARDS INDICATES END OF DIAGNOSTIC BEING WRITTEN.

00053	-0760	00	0	00141	SLT 1	IS THIS 2ND BLANK CARD
00054	0020	00	0	00060	TRA *+4	NO-TURN ON LITE 1
00055	0760	00	0	00142	SLN 2	YES-2ND BLANK CARD
00056	0760	00	0	00144	SLN 4	TURN ON LITE 4
00057	0020	00	0	00067	TRA AB	
00060	0760	00	0	00141	SLN 1	TURN ON LITE 1
00061	0020	00	0	00067	TRA AB	1ST BLANK CARD
00062	-0760	00	0	00141	AA SLT 1	WAS PREVIOUS CARD BLANK
00063	0020	00	0	00067	TRA AB	NO - PROCEED.
00064	0764	00	0	02221	{BRBB 1	YES - BLANK CARD IN THE
00065	0420	00	0	00000	{HPR	MIDDLE OF THE DECK RUN OF CARDS. 4TH CARD BACK WILL BE BLANK. REMOVE

BLANK AND RELOAD LAST 3
 CARDS AND THE REMAINDER OF
 THE DECK. PRESS START.

```

00066  000 00 0 00041      {TRA STRTA

*WRITE EACH CARD AS A RECORD ON CH B TAPE 1

00067  0774 00 4 00000  AB    AXT 0,4      CLEAR XRC
00070  0766 00 0 02221      WTBB 1      WRITE CARD IMAGE ON TAPE
00071  -0540 00 0 00305      RCHB CWA
00072  0061 00 0 00072      TCOB *
00073  -0022 00 0 00075      TRCB *+2    TRA ON TAPE CHECK
00074  0020 00 0 00102      TRA AC      OK
00075  0764 00 0 02221      {BRBB 1    BACKSPACE RECORD
00076  1 00001 4 00077      {TXI *+1,4,1  TRY AGAIN
00077  -3 00005 4 00101      {TXL *+2,4,5
00100  0420 00 0 00000      {HPR      TAPE WOULD NOT WRITE IN
00101  0020 00 0 00067      {TRA AB    5 TRIES - GET OFF MACHINE.

00102  0764 00 0 02221  AC    BRBB 1      BACKSPACE TAPE

00103  0760 00 0 00165      SWT 5      TEST SENSE SWITCH 5
00104  0020 00 0 00110      TRA RCDA   SELECT CARD READER
00105  -0760 00 0 00144      SLT 4      IS LITE 4 ON
00106  0020 00 0 00110      TRA RCDA   NO SELECT CARD READER
00107  0020 00 0 00112      TRA *+3    YES DO NOT SELECT
                                           CARD READER

*READ IN NEXT CARD WHILE CHECKING RECORD JUST WRITTEN

00110  0762 00 0 01321  RCDA  RCDA      READ NEXT CARD
00111  0540 00 0 00305      RCHA CWA

00112  0762 00 0 02221      RTBB 1      READ TAPE
00113  -0540 00 0 00306      RCHB CWB    FOR CHECKING
00114  0061 00 0 00114      TCOB *
00115  -0022 00 0 00117      TRCB *+2    TEST FOR TAPE CHECK
00116  0020 00 0 00122      TRA AD      OK
00117  000 00 0 00000      {HPR      TAPE CHECK ON READ.
                                           TO RESTART - RUN OUT CARDS
                                           AND RELOAD LAST 3 CARDS
                                           WITH REST OF DECK. THEN
                                           PRESS START TO TRY AGAIN.

00120  0764 00 0 02221      {BRBB 1
00121  0020 00 0 00041      {TRA STRTA

*COMPUTE CHECK SUM ON RECORD JUST WRITTEN

00122  0774 00 4 00030  AD    AXT 24,4    FORM CHECK SUM.
00123  -0754 00 0 00000      PXD      CLEAR ACC.
00124  0361 00 4 00367      ACL TSB+24,4
00125  2 00001 4 00124      TIX *-1,4,1
00126  0602 00 0 00370      SLW TSD    TAPE CHECK SUM

*COMPARE CHECK SUM OF RECORD JUST WRITTEN AGAINST CHECK SUM
  
```

*FROM CARD READER

00127	0500	00	0	00367		CLA TSC	CARD TO STG. CHECK SUM
00130	0402	00	0	00370		SUB TSD	COMPARE CHECK SUMS
00131	0100	00	0	00135		TZE AE	OK
00132	0420	00	0	00000		{HPR	CHECK SUM ERROR TO RESTART - RUN OUT CARDS AND RELOAD LAST 3 CARDS WITH REST OF DECK. THEN PRESS START TO TRY AGAIN
00133	000	00	0	00001		{BRBB 1	
00134	000	00	0	00041		{TRA STRTA	
00135	-0760	00	0	00142	AE	SLT 2	TEST FOR TAPE EOF DUE TO TWO BLANK CARDS.
00136	0020	00	0	00043		TRA STRTB	NO - CONTINUE
00137	0770	00	0	02221		WFBB 1	YES - WRITE END OF FILE AND
00140	0760	00	0	00165		SWT 5	
00141	0020	00	0	00043		TRA STRTB	CONTINUE
00142	0020	00	0	00254		TRA COUNT	WRITE NEW TAPE FROM CARDS AND TAPE
00143	0770	00	0	02201	EOF	WEFB 1	
00144	0760	00	0	00165		SWT 5	
00145	0020	00	0	00147		TRA *+2	WRITE NEW TAPE FROM CARDS
00146	0020	00	0	00254		TRA COUNT	WRITE NEW TAPE FROM CARDS AND TAPE
00147	0772	00	0	02201		REWB 1	
00150	0420	00	0	77777		HPR -1	CARD READER EOF
00151	0020	00	0	00150		TRA *-1	PROGRAM COMPLETED

*ROUTINE FOR MODIFICATION OF DIAGNOSTIC TAPE WITH SENSE SWITCH 5 ON

*RESET INDICATORS

00152	0772	00	0	02201	TAPE	REWB 1	
00153	0030	00	0	00154		TEFA *+1	RESET EOF INDICATOR
00154	-0030	00	0	00155		TEFB *+1	
00155	0022	00	0	00156		TRCA *+1	RESET REDUNDANCY INDICATOR
00156	-0022	00	0	00157		TRCB *+1	
00157	0420	00	0	00000		HPR	PUT TOTAL NUMBER OF FILES IN ADDRESS
00160	0760	00	0	00004		ENK	
00161	-0600	00	0	00302		STQ CTR	TOTAL NUMBER OF FILES ON TAPE
00162	0420	00	0	00000		HPR	ENTER KEY OR KEYS FOR FILE OR FILES TO BE READ FROM CARD READER

IF PROGRAM IS TO BE ADDED
TO TAPE. SET SIGN BIT AND
TOTAL NUMBER OF FILES ON
TAPE IN ADDRESS OF KEYS

00163	0760	00	0	00004		ENK	
00164	0131	00	0	00000		XCA	
00165	-0120	00	0	00265		TMI C	ADD PROGRAM TO REAR OF TAPE
00166	0772	00	0	01201		REWA 1	
00167	0767	00	0	00001	B1	ALS 1	
00170	0602	00	0	00371		SLW T	SAVE KEY
00171	-0760	00	0	00001		PBT	CHECK FOR P BIT
00172	0020	00	0	00201		TRA B2	READ FROM TAPE CHA AND WRITE ON CHB
*							
* SKIP OVER FILE BEING REPLACED BY NEW MODIFICATION ON TAPE 1 CH A							
*							
00173	0762	00	0	01221		RTBA 1	DO NOT TRANSMITT
00174	0540	00	0	00303		RCHA CW1	TO STG.
00175	0060	00	0	00175		TCOA *	
00176	0030	00	0	00200		TEFA *+2	TRANSFER ON EOF
00177	0000	00	0	00030		HTR START-3	NO EOF TRY AGAIN
							RELOAD CARDS
00200	0020	00	0	00034		TRA START+1	READ FROM CARD READER
*READ FROM CHA TAPE 1							
00201	0762	00	0	01221	B2	RTBA 1	READ FROM CH A
00202	0540	00	0	00305		RCHA CWA	
00203	0060	00	0	00203		TCOA *	
00204	0030	00	0	00253		TEFA B9	EOF ON TAPE 1 CHA
*COMPUTE CHECK SUM FROM CH A							
00205	0774	00	4	00030		AXT 24,4	COMPUTE CHECK SUM
00206	-0754	00	0	00000		PXD	CLEAR ACC
00207	0361	00	4	00337		ACL TSA+24,4	
00210	2	00001	4	00207		TIX *-1,4,1	
00211	0602	00	0	00367		SLW TSC	STORE CHECK SUM
00212	0022	00	0	00214		TRCA B3	TRA ON TAPE CHECK
00213	0020	00	0	00217		TRA B4	OK
00214	0420	00	0	00000	B3	HPR	ERROR REDUNDANCY CHECK
00215	-0764	00	0	01201		BSFA 1	BACKSPACE RECORD TRY AGAIN
00216	0020	00	0	00201		TRA B2	
*WRITE RECORD ON TAPE 1 CH B							
00217	0766	00	0	02221	B4	WTBB 1	WRITE RECORD JUST READ
00220	-0540	00	0	00305		RCHB CWA	
00221	0061	00	0	00221		TCOB *	
00222	-0022	00	0	00224		TRCB B5	ERROR ON WRITING
00223	0020	00	0	00227		TRA B6	OK
00224	0420	00	0	00000	B5	HPR	REDUNDANCY ERROR TRY AGAIN

00225	0764	00	0	02201		BSRB	1	
00226	0020	00	0	00217		TRA	B4	
00227	0764	00	0	02201	B6	BSRB	1	BACKSPACE RECORD
*CHECK RECORD JUST WRITTEN								
00230	0762	00	0	02221	B7	RTBB	1	
00231	000	00	0	00306		RCHAB	CWB	
00232	0061	00	0	00232		TCOB	*	
00233	-0022	00	0	00235		TRCB	*+2	TEST FOR TAPE CHECK
00234	0020	00	0	00237		TRA	B8	
00235	0420	00	0	00000		HPR		REDUNDANCY ERROR BST + TRY AGAIN
00236	0020	00	0	00227		TRA	B6	
* COMPUTE CHECK SUM ON RECORD JUST WRITTEN								
00237	0774	00	4	00030	B8	AXT	24,4	COMPUTE CHECK SUM
00240	-0754	00	0	00000		PXD		CLEAR ACC.
00241	0361	00	4	00367		ACL	TSB+24,4	
00242	2	00001	4	00241		TIX	*-1,4,1	
00243	0602	00	0	00370		SLW	TSD	
* COMPARE CHECK SUM WRITTEN AGAINST SUM READ								
00244	0500	00	0	00367		CLA	TSC	
00245	0402	00	0	00370		SUB	TSD	
00246	0100	00	0	00201		TZE	B2	OK
00247	0420	00	0	00000		HPR		ERROR ON CHECK SUM
00250	0764	00	0	01201		BSRA	1	TRY READING AND WRITING
00251	0764	00	0	02201		BSRB	1	AGAIN
00252	0020	00	0	00201		TRA	B2	
00253	0770	00	0	02201	B9	WEFB	1	WRITE EOF OF FILE ON NEW TAPE
*REDUCE COUNTER CONTAINING NUMBER OF FILES ON TAPE								
00254	0500	00	0	00302	COUNT	CLA	CTR	NO OF FILES
00255	0402	00	0	00301		SUB	PONE	L +1
00256	0601	00	0	00302		STO	CTR	
00257	0100	00	0	00262		TZE	REW	LAST PROG. ON TAPE
00260	-0500	00	0	00371		CAL	T	
00261	0020	00	0	00167		TRA	B1	READ NEXT FILE
00262	0772	00	0	01201	REW	REWA	1	
00263	0772	00	0	02201		REWB	1	
00264	0000	00	0	00030		HTR	24	FINISHED NEW TAPE COMPLETED
*ROUTINE TO ADD NEW PROGRAM ON END OF TAPE								
00265	0621	00	0	00302	C	STA	CTR	
00266	0762	00	0	02221	C1	RTBB	1	DO NOT TRANSMITT TO STG.
00267	-0540	00	0	00303		RCHB	CW1	

00270	0061	00	0	00270		TCOB *	
00271	-0030	00	0	00273		TEFB C2	
00272	0000	00	0	00030		HTR 24	NO EOF TRY AGAIN RELOAD CARDS
00273	0500	00	0	00302	C2	CLA CTR	
00274	0402	00	0	00301		SUB PONE	
00275	0601	00	0	00302		STO CTR	
00276	0100	00	0	00300		TZE *+2	
00277	0020	00	0	00266		TRA C1	
00300	0000	00	0	00034		HTR START+1	HALT TO PUT SWT 5 UP. THIS WILL ALLOW FULL USE OF CARD READER
00301	+00000000000001				PONE	OCT 1	
00302	+00000000000000				CTR	OCT 0	
00303	-0777 77 2	00000			CW1	IOCP 0,2,-1	
00304	1 00000	0	00303			TCH *-1	
00305	0000 30	0	00307	CWA		IOCD TSA,0,24	
00306	0000 30	0	00337	CWB		IOCD TSB,0,24	
			00307	TSA		BSS 24	
			00337	TSB		BSS 24	
			00367	TSC		BSS 1	
			00370	TSD		BSS 1	
			00371	T		BSS 1	
			00000			END	

EOF*

TAPE REWINDER FOR LOW END LOADER 9LD01

LOAD THIS PROGRAM IN FRONT OF THE
LAST TRANSFER CARD OF EACH
FILE OF CARDS TO GO ON SYSTEM
TAPE. THIS VERSION OF THE
REWINDER IS MATCHED
TO LOADER 9LD01 WITH TAPE SELECTION.
BE SURE THAT LOADER AND REWINDER
ARE MATCHED.

				00004		ORG 4		
00004	0762	00	0	01221	A	RTBA 1	READ	TRANSFER CARD
00005	0540	00	0	00025		RCHA B		
00006	0060	00	0	00006		TCOA *		
00007	0772	00	0	01201		REWA 1	REWIND	TAPE 1
00010	0500	00	0	00026		CLA B+1		
00011	0100	00	0	00027		TZE B+2	TRANSFER	TO MAIN PROGRAM
00012	0000	00	0	00012		HTR *	BAD	TRANSFER CARD
				00025	B	EQU 21		
				00000		END		

TAPE REWINDER FOR HIGH END LOADER 9LD02

LOAD THIS PROGRAM IN FRONT OF THE
LAST TRANSFER CARD OF EACH
FILE OF CARDS TO GO ON SYSTEM
TAPE. THIS VERSION OF THE
REWINDER IS MATCHED
TO LOADER 9LD02 WITH TAPE SELECTION.
BE SURE THAT LOADER AND REWINDER
ARE MATCHED.

```

                                77753          ORG 32747
77753  0762 00 0 01221  A      RTBA 1          READ TRANSFER CARD
77754  0540 00 0 77774          RCHA B
77755  0060 00 0 77755          TCOA *
77756  0772 00 0 01201          REWA 1          REWIND TAPE 1
77757  0500 00 0 77775          CLA B+1
77760  0100 00 0 77776          TZE B+2          TRANSFER TO MAIN PROGRAM
77761  0000 00 0 77777          HTR -1          BAD TRANSFER CARD

                                77774  B      EQU -4
                                00000          END
```

EOF*